Publication Number C29-529R02

EXTENDED SELECTOR CHANNEL (ESELCH)

PROGRAMMING MANUAL

PERKIN-ELMER

Computer Systems Division
2 Crescent Place
Oceanport, N.J. 07757

Copyright © 1976 by Perkin-Elmer Corporation Printed in U.S.A. May 1980

PAGE REVISION STATUS SHEET

PUBLICATION NUMBER C29-529

TITLE Extended Selector Channel (ESELCH) Programming Manual

REVISION R02 DATE May 1980

PAGE	REV.	DATE	PAGE	REV.	DATE	PAGE	REV.	DATE
i/ii	R01	3/77						
iii/iv	R01	3/77						
1	R01	3/77						
2	R02	5/80						
3	R00	5/76						
4	R00	5/76						
5	R00	5/76						
6	R00	5/76						
7/8	R00	5/76						
A1-1	R00	5/76						
A1-2	R00	5/76						
A1-3	R00	5/76						
A1-4	R00	5/76						
A1-5	R00	5/76						
A1-6	R00	5/76						
A1-7	R00	5/76						
A1-8	R00	5/76						
A1-9	R00	5/76						
A1-10	R00	5/76						
A1-11/ A1-12	R00	5/76						
I-1	R01	3/77						

A2254

PREFACE

This manual provides the necessary information to program the ESELCH. It contains sections on Configuration Programming Instructions, Status and Command Byte information, Programming Sequences, Programming Notes, Device Number, and Initialization. A programming example is provided in Appendix 1. A detailed index is provided at the rear of this manual.

Table of Contents

ILLUSTRATIONS

TABLES

APPENDIX

EXTENDED SELECTOR CHANNEL (ESELCH) PROGRAMMING MANUAL

INTRODUCTION

The M73-105 ESELCH and the M48-050 BSELCH are both upward compatible with the 16-bit address Selector Channel. They control data transfers between I/O devices and local or extended memories at rates up to 2,000,000 bytes per second in the Halfword Mode. The BSELCH can transfer data up to six megabytes per second in the Burst Mode. A maximum of 16 I/O devices can be connected to the SELCHs, but only one device at a time can transfer data. Data transfers can be either byte or halfword oriented depending on the particular I/O device.

This manual is applicable to both the ESELCH and BSELCH, however, the text only references the BSELCH for ease of reading.

CONFIGURATION

Refer to the Extended Selector Channel Installation Specification 02-328A20.

PROGRAMMING INSTRUCTIONS

A Sense Status instruction (SS or SSR) is used to transfer the ESELCH Status Byte to the processor. Refer to Table 1 for Status and Command Byte information. This instruction should not be used under interrupt control for it could cause the ESELCH to become idle and the interrupt condition to become reset.

The Output Command instruction (OC or OCR) causes a Command Byte to be sent to the ESELCH.

The Write Data (WD or WDR) or Write Halfword (WH or WHR) instructions are used to send the starting and final addresses to the ESELCH.

The Read Data (RD or RDR) or Read Halfword (RH or RHR) instructions are used to obtain the last processor memory location either written into or read from memory.

The Write Block (WB or WBR) instruction or Read Block (RB or RBR) instruction should not be used.

TABLE 1. ESELCH STATUS AND COMMAND BYTE DATA

BIT NUMBER	8	9	10	11	12	13	14	15
STATUS BYTE			MEMORY MALFUNC-TION	MEMORY PARITY FAIL	BUSY			
COMMAND BYTE		EXTENDED ADDRESS READ	READ	GO	STOP	SELCH STATUS		

STATUS BYTE DESCRIPTION

Busy — This bit is set if the ESELCH is transferring data and reset when idle or not transferring data.

Memory Parity Fail — This bit is set if the memory interface recognizes a parity failure and is available for subsequent evaluation by the processor. However, the transfer is not interrupted. Resetting this bit is by processor initialization or by the Output Command Go.

Memory Malfunction — This bit is set if the memory interface recognizes a malfunction other than parity failure and is available for subsequent processor evaluation. Transfer is not interrupted. Resetting this bit is by processor initialization or by the Output Command Go.

COMMAND BYTE DESCRIPTION

Extended Address Read — This command specifies whether two or three final address bytes are returned to the processor. If set, three bytes are returned, and if reset, two bytes are returned. This command is issued concurrently when specifying the Stop mode.

Read — This command changes the ESELCH mode from Write to Read. In the Read mode, data is transmitted from the active ESELCH device and written into memory. The ESELCH goes to the Write mode at the completion of a data transfer. Thus, each time a Read operation is required, a Read command is issued to the ESELCH.

Go — This command initiates data transmission and can be issued when specifying a Read or Write mode.

Stop — This command halts a data transmission and initializes the ESELCH for starting a new operation. This command should be issued when the ESELCH terminates.

NOTE

A Stop command issued to a Busy SELCH will not set the MSC1 flip-flop, therefore not allowing communication to devices on the SELCH Private Bus. The SELCH must be addressed a second time to allow communications under the above conditions.

SELCH Status — If this bit is set, the ESELCH status is returned by SS or SSR instructions to the ESELCH. If reset, and the ESELCH is Busy, only the Busy bit is present in the Status Byte. All other bits are ZERO. If reset, and the ESELCH is not Busy, the device status is present in the Status Byte with the Busy bit forced to ZERO.

PROGRAMMING SEQUENCES

Programming a device on the ESELCH consists of setting up the device, setting up the ESELCH, and sending a Go command to the ESELCH as explained in the section entitled Programming Instructions. Setting up or initializing a device on the ESELCH is device dependent, and the user should refer to the appropriate device programming manual for specifics. If the ESELCH is Busy, I/O instructions should not be issued, but Stop, Extended Address Read, or SELCH Status commands can be issued. Like all other devices, the ESELCH has its own unique device number and is affected by Output Commands, as with all I/O instructions, only if the correct device address is used. If a device on the ESELCH is referenced while the ESELCH is Busy, the False SYNC bit of the Condition Code is set (CVGL = 0100) indicating an ignored command.

The ESELCH is idle (not Busy) only after an Initialize or Stop command is issued to the ESELCH. Thus, prior to issuing an I/O operation with the ESELCH or any device on the ESELCH bus, the program must do one of the following:

1. Wait until the ESELCH completes an I/O transfer and then issue a Stop command to the ESELCH.
2. Issue a Stop command to the ESELCH.

In order to perform a Read or Write operation on a device using an ESELCH, the following steps are required:

1. Wait for the ESELCH to become not Busy, or issue a Stop command. Note that use of the Stop command will stop any transfer in progress, and should be used wisely.
2. Set up the device on the ESELCH by issuing the appropriate I/O commands.
3. Send the starting and ending address to the ESELCH.
4. Send an Output Command to the ESELCH with the Go bit set. The Read bit should be set for a Read operation, and reset for a Write operation.

NOTE

No I/O instruction can be issued to any other device during the execution of the program instructions in Steps 2 through 4. When the Go command is issued, the ESELCH does the following:

1. Controls the last device addressed by the processor (i.e., by the user program or by the microprogram) if that device is connected to the ESELCH.
2. Hangs if no device attached to the ESELCH has been addressed by the processor since the last processor Initialization. Refer to the section entitled INITIALIZATION.

For this reason, the program instructions in Steps 2 through 4 must be executed with external interrupts disabled (PSW Bit 17 = 0). Otherwise, it is possible to detect an external interrupt and the interrupt driver in the user program (or the microprogram in the case of an immediate interrupt) can address any other device. Also, the single mode may not be used since the display panel is addressed in this mode.

PROGRAMMING NOTES

The ESELCH has a 20-bit incrementing Address Register and a 20-bit Final Address Register. The user program loads the starting address into the incrementing Address Register and the final address into the Final Address Register. Transfer is completed when the incrementing Address Register matches the Final Address Register. For additional information on the operation of these registers, refer to the ESELCH Maintenance Specification, 02-328A21.

Memory is addressed on halfword boundaries; that is, each time memory is accessed, two bytes or a halfword are obtained. Referring to Figure 1, 20-bit addresses are used in the memory system, with the least significant bit, Bit 19, specifying the desired byte.



Figure 1. Memory Addressing

Each time the ESELCH accesses memory, two bytes (one halfword) are transferred. *It is mandatory that data transfers begin on a halfword boundary.*

The following results if data transfers are ended on byte boundaries:

1. Write mode (memory to device) – End on byte boundary (Bit 19 = 0). No effect, the transfer can terminate successfully.

2. Read mode (device to memory) – End on byte boundary (Bit 19 = 0). The previous contents of the last odd byte in memory are written into the current odd byte in memory. Refer to Figure 2. The user should ignore the last (unrequested) byte.

Data transfers across a 256KB boundary may be executed provided the memory is contiguous.



Figure 2. Core Memory Configuration for Ending on Byte Boundary

Transmission of Starting and Final Addresses

An Output Command with the Stop bit set should be issued prior to starting an operation on the ESELCH to clear any preceding conditions. Normally, six successive bytes are required to specify starting and final addresses of the user's buffer area. If the final address of the buffer area is no greater than 65536, either four or six (with the most significant byte zero) successive bytes may be used to specify the starting and final addresses. Figure 3 illustrates the meaning of the six bytes used for addressing.



* MUST BE ZERO

1. STARTING ADDRESS (BITS 0-3)
2. STARTING ADDRESS (BITS 4-11)
3. STARTING ADDRESS (BITS 12-19)
4. FINAL ADDRESS (BITS 0-3)
5. FINAL ADDRESS (BITS 4-11)
6. FINAL ADDRESS (BITS 12-19)

Figure 3. Meaning of Data Bytes for Setting Start and Final Addresses

Either the Write Data (WD or WDR) or Write Halfword (WH or WHR) instructions may be used to send the starting and final addresses to the ESELCH.

For example, if the starting address is X'000000' and the ending address is X'ABCDE', the following addressing sequences are all correct:

1. WD (or WDR) SELCH, LARX (00)
   WD (or WDR) SELCH, LARH (00)
   WD (or WDR) SELCH, LARL (00)
   WD (or WDR) SELCH, LFAX (0A)
   WD (or WDR) SELCH, LFAH (BC)
   WD (or WDR) SELCH, LFAL (DE)

2. WD (or WDR) SELCH, LARX (00)
   WH (or WHR) SELCH, LARHA (0000)
   WD (or WDR) SELCH, LFAX (0A)
   WH (or WHR) SELCH, LFAHL (BCDE)

3. WH (or WHR) SELCH, LARXH (0000)
   WD (or WDR) SELCH, LARL (00)
   WH (or WHR) SELCH, LRAXH (0ABC)
   WD (or WDR) SELCH, LFAL (DE)

If the starting address is X'001234' and the ending address is X'00ABCD', the following addressing sequences are all correct:

1. WD (or WDR) SELCH, LARH (12)
   WD (or WDR) SELCH, LARL (34)
   WD (or WDR) SELCH, LFAH (AB)
   WD (or WDR) SELCH, LFAL (CD)

2. WH (or WHR) SELCH, LARHL (1234)
   WH (or WHR) SELCH, LAFAHL (ABCD)

3. WD (or WDR) SELCH, LARX (00)
   WD (or WDR) SELCH, LARH (12)
   WD (or WDR) SELCH, LARL (34)
   WD (or WDR) SELCH, LFAX (00)
   WD (or WDR) SELCH, LFAH (AB)
   WD (or WDR) SELCH, FLAL (CD)

4. WD (or WDR) SELCH, LARX (00)
   WH (or WHR) SELCH, LARHL (1234)
   WD (or WDR) SELCH, LFAX (00)
   WH (or WHR) SELCH, FLAHL (ABCD)

5. WH (or WHR) SELCH, LARXH (0012)
   WD (or WDR) SELCH, LARL (34)
   WH (or WHR) SELCH, LFAXH (00AB)
   WD (or WDR) SELCH, LFAL (CD)

Termination

Data transmission between the ESELCH and the device presently connected to it is halted if any of the following conditions occur:

1. The starting address matches the final address. This denotes normal termination.

2. The starting (incrementing) address goes from all ones to all zeros (maximum count). In this case, a match has not occurred and is considered an abnormal termination.

3. Any of DU, EOM, or EX status bits of the device presently connected to the ESELCH changes to a ONE. This may be an abnormal termination, depending on the application.

4. A Stop command is sent to the ESELCH via a user program. The termination condition is determined in one of two ways – by a status scan, or by the interrupt method. An Output Command Stop should be issued to the ESELCH following its termination.

5. Addressing a non-existent memory.

Status Scan

The status of the ESELCH Controller may be examined by issuing a Sense Status (SS or SSR) instruction. The Busy bit (Bit 12) is a ONE while transmission is in progress, and ZERO when transmission is terminated. One method of testing for termination is to continually or periodically test the Busy bit of the ESELCH. The change from ONE to ZERO indicates the termination of a data transfer. In the status scan method of programming, it is possible for the Busy bit to change from ONE to ZERO during a Sense Status instruction without returning the ESELCH to idle. To guarantee the idle mode after Busy = 0 on a Sense Status instruction, a Stop command must be sent to the ESELCH.

Interrupt Method

When data transmission is initiated on the ESELCH, the interrupt is always enabled. If external device interrupts are enabled, the processor is interrupted when the ESELCH terminates. The interrupt causes program control to transfer to the start address of a user's termination service routine via a pointer located at an Interrupt Service Pointer Table address by the formula: 2 X device Address + X'D0'. To guarantee an idle mode after the ESELCH interrupt, a Stop command must be sent to the ESELCH.

Reading the Final Address

The last processor memory location either written into or read from may be obtained by executing a pair of Read Data (RD or RDR) instructions or a Read Halfword (RH or RHR) instruction provided the final address is no greater than 65536. Three successive Read Data (RD or RDR) instructions or one Read Data and one Read Halfword instruction are required if the final address is greater than 65536.

The command Extended Address Read (Bit 9) specifies whether a 2 byte or 3 byte final address is used. Before issuing RD or RH instruction, a Stop command should be issued to insure that the ESELCH is initialized.

This information permits a user program to verify a successful data transmission or to determine what address termination occurred.

Table 2 illustrates various sequences in which the final address can be read.

TABLE 2. SEQUENCES USED TO READ FINAL ADDRESS

(A) FINAL ADDRESS IS X'00ABCD'

INSTRUCTION SEQUENCE	ANSWERS	
	BIT 9 SET	BIT 9 RESET
RD (OR RDR) RD (OR RDR)	00 AB	AB CD
RH (OR RHR)	00AB	ABCD
RD (OR RDR) RD (OR RDR) RD (OR RDR)	00 AB CD	AB CD UNDEFINED
RD (OR RDR) RH (OR RHR)	00 ABCD	AB CDXX
RH (OR RHR) RD (OR RDR)	00AB CD	ABCD UNDEFINED

(B) FINAL ADDRESS IS X'0ABCDE'

INSTRUCTION SEQUENCE	ANSWERS	
	BIT 9 SET	BIT 9 RESET
RD (OR RDR) RD (OR RDR)	0A BC	BC DE
RH (OR RHR)	0ABC	BCDE
RD (OR RDR) RD (OR RDR) RD (OR RDR)	0A BC DE	BC DE UNDEFINED
RD (OR RDR) RH (OR RHR)	0A BCDE	BC DEXX
RH (OR RHR) RD (OR RDR)	0ABC DE	BCDE UNDEFINED

DEVICE NUMBER

The ESELCH is normally assigned device number X'0F0' (10-bit address), but may be easily changed on the ESELCH Device Controller Board. Refer to Installation Specification 02-328A20 for details.

INITIALIZATION

Whenever the Initialize switch (INT) on the Display Panel is depressed, or a Stop Command is issued, the following action occurs:

1. Any data transmission in process is halted and the Stop Mode is affected.
2. The ESELCH is placed in the Write Mode.
3. The ESELCH is made idle.
4. The ESELCH interrupt is reset.

APPENDIX 1
PROGRAMMING EXAMPLE



* REFER TO A1-2
FOR DRIVER ROUTINE


C
READ FINAL ADDRESS FROM ESELCH
READ ADRS = EXPECTED ADRS ?
NO
HALT
YES
INPUT -OUTPUT BUFFERS MATCH ?
NO
HALT
YES
DISPLAY X'FFFFFFFF' ON CONSOLE
DONE PART 2
DRIVER
DISC CONTROLLER IDLE ?
NO
CONTROL FALSE SYNC ?
NO
YES
HALT
YES
RESTORE DISC FILE
DISC FILE STATUS O.K. ?
NO
HALT
YES
LOAD DISC FILE WITH CYLINDER NUMBER
SEEK
FILE READY TO SEEK, READ, OR WRITE ?
NO
DEPENDING ON THE DRIVER IMPLEMENTATION METHOD, IT MAY BE REQUIRED TO WAIT FOR THE FRSRW INTERRUPT.
YES
LOAD CYLINDER NO. TO DISC FILE AND LOAD HEAD AND SECTOR NO. TO CONTROLLER
RETURN TO CALLING RT.

```
     ESELCH PROGRAMMING EXAMPLE                                PAGE    1    14:41:49  03/22/76

PROG= *NONE*      ASSEMBLED BY CAL 03-066R04(32-BIT)

                                        1           SCRAT
                                        2           TARGT  32
                                        3           NORX3
                                        4           WIDTH  120
                                        5           CROSS
 000000I                                6           ORG    *
          0000 0000                     7  R0       EQU    0
          0000 0001                     8  R1       EQU    1
          0000 0002                     9  R2       EQU    2
          0000 0003                    10  R3       EQU    3
          0000 0004                    11  SELCH    EQU    4
          0000 0005                    12  CONTER   EQU    5
          0000 0006                    13  DISFIL   EQU    6
          0000 0007                    14  SESTAT   EQU    7
          0000 0008                    15  COSTAT   EQU    8
          0000 0009                    16  DISTAT   EQU    9
          0000 000A                    17  WORKA    EQU    10
          0000 000B                    18  WORKB    EQU    11
          0000 000C                    19  WORKC    EQU    12
          0000 000D                    20  RTN      EQU    13
          0000 000E                    21  R14      EQU    14
          0000 000F                    22  R15      EQU    15
```

ESELCH PROGRAMMING EXAMPLE PAGE 2 14:41:53 03/22/76

```
                                    24  *
                                    25  *
                                    26  *       IN THIS PORTION OF THE EXAMPLE, ESELCH IS UNDER STATUS CONTROL
                                    27  *       I/O DEVICE IS A DISC CONTROLLER ADDRESSED AT X'B6', THE
                                    28  *       ASSOCIATED DISC FILE IS ADDRESSED AT X'C6'.
                                    29  *       THE DATA IN THE OUTPUT BUFFER IS WRITTEN TO THE DISC FILE
                                    30  *       THROUGH THE ESELCH.
                                    31  *
                                    32  *
                                    33  *
000000I  C810 20F0                  34  START   LHI     R1,X'20F0'          DISARM ANY EXTERNAL INTERRUPT
000004I  9531                       35          EPSR    R3,R1
000006I  7340 8254                  36          LHL     SELCH,SELADR        LOAD SELCH ADDRESS
00000AI  DE40 826A                  37          OC      SELCH,STOP1         STOP ESELCH BEFORE ANY OPERATION
00000EI  4240 8242                  38          B0      HALT                HALT IF ESELCH FALSE SYNC
000012I  9D47                       39  SENSE   SSR     SELCH,SESTAT
000014I  2081                       40          BCS     SENSE               WAIT UNTIL BUSY BIT = 0
000016I  41D0 81B4                  41          BAL     RTN,DRIVER          SET UP IO DEVICE
00001AI  E6A0 8426                  42          LA      WORKA,OUTBUF
00001EI  F8B0 0000 0544I            43          LI      WORKB,ENDOUT
000024I  34AA                       44          EXHR    WORKA,WORKA
000026I  9A4A                       45          WDR     SELCH,WORKA         WRITE FIRST BYTE OF STARTING ADDRESS
000028I  34AA                       46          EXHR    WORKA,WORKA
00002AI  9846                       47          WHR     SELCH,WORKA         WRITE SECOND AND THRID BYTES
00002CI  34BB                       48          EXHR    WORKB,WORKB
00002EI  9A4B                       49          WDR     SELCH,WORKB         WRITE FIRST BYTE OF FINAL ADDRESS
000030I  23BB                       50          EXHR    WORKB,WORKB
000032I  984B                       51          WHR     SELCH,WORKB         WRITE SECOND AND THIRD BYTES
000034I  7350 8228                  52          LHL     CONTER,CONADR       LOAD CONTROLLER ADDRESS
000038I  DE50 8241                  53          OC      CONTER,DWRITE       ISSUE WRITE COMMAND
00003CI  DE40 823B                  54          OC      SELCH,WRTCMD        WRITE MODE, GO
                                    55  *
                                    56  *       WRITE OPERATION UNDER STATUS CONTROL
000040I  9D47                       57  WAITLP  SSR     SELCH,SESTAT        WAIT UNTIL TRANSFER IS COMPLETED
000042I  2081                       58          BCS     WAITLP
000044I  DE40 8230                  59          OC      SELCH,STOP1         ENSURE STOP
000048I  9047                       60          SSR     SELCH,SESTAT
00004AI  C370 0030                  61          THI     SESTAT,X'30'        MEMORY FAILURE, PARITY, ERROR ?
00004EI  4230 8202                  62          BNZ     HALT                ABNORMAL TERMINATE
000052I  9B4A                       63          RDR     SELCH,WORKA         CHECK THE FINAL ADDRESS
000054I  08CA                       64          LR      WORKC,WORKA         FIRST BYTE
000056I  994A                       65          RHR     SELCH,WORKA         SECOND AND THIRD BYTES
000058I  34CC                       66          EXHR    WORKC,WORKC
00005AI  06CA                       67          OR      WORKC,WORKA
00005CI  05BC                       68          CLR     WORKB,WORKC         COMPARE READ AND EXPECTED ADDRESSES
00005EI  4230 81F2                  69          BNE     HALT                HALT IF NOT EQUAL
000062I  DE40 8213                  70          OC      SELCH,STOP2         SELCH STATUS BIT RESET
000066I  9D48                       71          SSR     SELCH,COSTAT        SENSE THE CONTROLLER STATUS
000068I  C380 00C1                  72          THI     COSTAT,X'C1'        ANYTHING WRONG ?
00006CI  4230 81E4                  73          BNZ     HALT
```

```
                                    75  *
                                    76  *
                                    77  *       IN THIS PORTION OF THE EXAMPLE, ESELCH IS UNDER INTERRUPT
                                    78  *       CONTROL.  THE PROGRAM READS THE WRITTEN DATA FROM THE DISC FILE
                                    79  *       AND LOADS IT INTO INPUT BUFFER
                                    80  *       AFTER THE DATA HAS BEEN READ CORRECTLY, THE INPUT AND OUTPUT
                                    81  *       BUFFERS ARE COMPARED, IF THEY ARE MATCHED, THE PATTERN
                                    82  *       X'FFFFFFFF' IS DISPLAYED ON THE CONSOLE.
                                    83  *
                                    84  *
                                    85  *
        0000 0070I                  86  START?  EQU     *
000070I 07AA                        87          XR      WORKA,WORKA
000072I 40AC 81FA                   88          STH     WORKA,FLAG
000076I E6A0 810C                   89          LA      WORKA,SELINT
00007AI 7380 81E0                   90          LHL     WORKB,SELADR
00007EI 11B1                        91          SLLS    WORKB,1             SELCH ADDRESS X 2
000080I 40AB 00D0                   92          STH     WORKA,X'D0'(WORKB)
000084I F6A0 8116                   93          LA      WORKA,COINT
000088I 73B0 81D4                   94          LHL     WORKB,CONADR
00008CI 11B1                        95          SLLS    WORKB,1             CONTROLLER ADDRESS X 2
00008EI 40AB 00D0                   96          STH     WORKA,X'D0'(WORKB)
000092I E6A0 8120                   97          LA      WORKA,DISINT
000096I 73B0 81C8                   98          LHL     WORKB,FILADR
00009AI 11B1                        99          SLLS    WORKB,1             DISC FIL ADDRESS X 2
00009CI 40AB 00D0                  100          STH     WORKA,X'D0'(WORKB)
0000A0I DE40 81D4                  101          OC      SELCH,STOP1
0000A4I 4240 81AC                  102          BO      HALT                ABORT IF FALSE SYNC
0000A8I 41D0 8122                  103          BAL     RTN,DRIVER          SET UP I/O DEVICE
0000ACI E6A0 8294                  104          LA      WORKA,INBUF
0000B0I F8B0 0000 0444I            105          LI      WORKB,ENDINB
0000B6I 34AA                       106          EXHR    WORKA,WORKA
0000B8I 9A4A                       107          WDR     SELCH,WORKA         WRITE FIRST BYTE
0000BAI 34AA                       108          EXHR    WORKA,WORKA
0000BCI 984A                       109          WHR     SELCH,WORKA         WRITE SECOND AND THIRD BYTES
0000BEI 34BB                       110          EXHR    WORKB,WORKB
0000C0I 9A4B                       111          WDR     SELCH,WORKB
0000C2I 34BB                       112          EXHR    WORKB,WORKB
0000C4I 984B                       113          WHR     SELCH,WORKB
                                   114  *
                                   115  *       READ OPERATION UNDER INTERRUPT CONTROL
0000C6I DE50 81B2                  116          OC      CONTER,DREAD        ISSURE READ COMMAND TO CONTROLLER
0000CAI DE40 81AC                  117          OC      SELCH,RDCMD         READ MODE, GO
0000CEI C8A0 60F0                  118          LHI     WORKA,X'60F0'       ENABLE ALL INTERRUPT LEVEL
0000D2I 95BA                       119          EPSR    WORKB,WORKA
0000D4I 73A0 8198                  120  BACKGR  LHL     WORKA,FLAG          WAIT UNTIL INTERRUPTS GENERATED
0000D8I 27A2                       121          SIS     WORKA,2
0000DAI 2013                       122          BMS     BACKGR
                                   123  *       AT THIS POINT EXPECTED INTERRUPTS HAVE BEEN GENERATED
0000DCI C8A0 20F0                  124          LHI     WORKA,X'20F0'
0000E0I 95BA                       125          EPSR    WORKB,WORKA         DISABLE ALL EXTERNAL INTERRUPT
0000E2I 73A0 817E                  126          LHL     WORKA,SELCHX
0000E6I 054A                       127          CLR     SELCH,WORKA
0000E8I 4230 8168                  128          BNZ     HALT                INCORRECT SELCH INTERRUPT ADDRESS
```

ESELCH PROGRAMMING EXAMPLE PAGE 4 14:42:06 03/22/76

```
0000FCI  73A0 8176          129          LHL    WORKA,SELSTX
0000F0I  4230 8160          130          BNZ    HALT              ABNORMAL ESELCH TERMINATE
0000F4I  73A0 8170          131          LHL    WORKA,CONADRX
0000F8I  055A               132          CLR    CONTER,WORKA
0000FAI  4230 8156          133          BNZ    HALT              INCORRECT CONTROLLER ADDRESS
0000FEI  7380 8168          134          LHL    COSTAT,CONSTX
000102I  C380 00C1          135          THI    COSTAT,X'C1'      ANY ERROR IN THE STATUS BYTE?
000106I  4230 814A          136          BNZ    HALT
00010AI  73A0 815E          137          LHL    WORKA,FILADX      DISC FILE
00010EI  4330 8012          138          BZ     FINCHK
000112I  05A6               139          CLR    WORKA,DISFIL
000114I  4230 813C          140          BNZ    HALT              INCORRECT FILE ADDRESS
000118I  7390 8152          141          LHL    DISTAT,FILSTX
00011CI  C390 0021          142          THI    DISTAT,X'21'
000120I  4230 8130          143          BNZ    HALT              DISC FILE ERROR
000124I  DE40 8150          144  FINCHK  OC     SELCH,STOP1
000128I  4240 8128          145          BO     HALT
00012CI  9B4A               146          RDR    SELCH,WORKA       FINAL ADDRESS FIRST BYTE
00012EI  08CA               147          LR     WORKC,WORKA
000130I  994A               148          RHR    SELCH,WORKA       SECOND AND THIRD BYTES
000132I  34CC               149          EXHR   WORKC,WORKC
000134I  06CA               150          OR     WORKC,WORKA       FINAL ADDRESS FROM READING
000136I  F8B0 0000 0444I    151          LI     WORKB,ENDINB      EXPECTED FINAL ADDRESS
00013CI  05BC               152          CLR    WORKB,WORKC
00013EI  4230 8112          153          RNZ    HALT              ABNORMAL TERMINATE
                            154  *       CHECK INPUT AND OUTPUT BUFFERS
000142I  E620 81FE          155          LA     R2,INBUF
000146I  E630 82FA          156          LA     R3,OUTBUF
00014AI  C8C0 007E          157          LHI    WORKC,126
00014EI  24B2               158          LIS    WORKB,2
000150I  07AA               159          XR     WORKA,WORKA
000152I  7312 4A00 0000     160  MATCH   LHL    R1,0(R2,WORKA)
000158I  7303 4A00 0000     161          LHL    R0,0(R3,WORKA)
00015EI  0501               162          CLR    R0,R1
000160I  4230 80F0          163          BNZ    HALT              READ AND WRITE DATA NOT MATCH
000164I  C1A0 FFEA          164          BXLE   WORKA,MATCH
000168I  C8B0 FFFF          165  DONE    LHI    WORKB,X'FFFF'     WORKB=FFFFFFFF
00016CI  24A1               166          LIS    WORKA,1
00016EI  DEA0 810D          167          OC     WORKA,INC
000172I  98AB               168          WHR    WORKA,WORKB
000174I  98AB               169          WHR    WORKA,WORKB
000176I  0ACC               170          AR     WORKC,WORKC
000178I  9EAC               171          OCR    WORKA,WORKC
00017AI  F8A0 0000 80F0     172          LI     WORKA,X'80F0'     CONSOLE IN WAIT MODE
000180I  95BA               173          EPSR   WORKB,WORKA
000182I  4300 FE7A          174          B      START             REPEAT THE ROUTINE IF DESIRED
                            175  *
                            176  *
                            177  *       ESELCH INTERRUPT HANDLE ROUTINE
                            178  *
000186I  D000 80FA          179  SELINT  STM    R0,RSAVE          SAVE ALL REG
00018AI  4020 80D6          180          STH    R2,SELCHX         STORE SELCH ADDRESS AT INTERRUPTED
00018EI  4030 80D4          181          STH    R3,SELSTX         STORE SELCH STATUS AT INTERRUPTED
000192I  24A1               182          LIS    WORKA,1
```

ESELCH PROGRAMMING EXAMPLE PAGE 5 14:42:13 03/22/76

```
000194I  61A0 80D8          183           AHM    WORKA,FLAG
000198I  D100 80E8          184           LM     R0,RSAVE          RESTOR ALL REG
00019CI  1800               185           LPSWR  R0                RETURN TO INTERRUPTED LOCATION
                            186  *
                            187  *
                            188  *        CONTROLLER INTERRUPT HANDLE ROUTINE
                            189  *
00019EI  D000 8122          190  COINT    STM    R0,RSAVE1
0001A2I  4020 80C2          191           STH    R2,CONADRX        STORE INTERRUPTED ADDRESS
0001A6I  4030 80C0          192           STH    R3,CONSTX         STORE STATUS
0001AAI  24A1               193           LIS    WORKA,1
0001ACI  61A0 80C0          194           AHM    WORKA,FLAG
0001B0I  D100 8110          195           LM     R0,RSAVE1
0001B4I  1800               196           LPSWR  R0
                            197  *
                            198  *
                            199  *        DISC FILE INTERRUPT HANDLE ROUTINE
                            200  *
0001B6I  D000 814A          201  DISINT   STM    R0,RSAVE2
0001BAI  4020 80AE          202           STH    R2,FILADX
0001BEI  4030 80AC          203           STH    R3,FILSTX
0001C2I  24A1               204           LIS    WORKA,1
0001C4I  61A0 80A8          205           AHM    WORKA,FLAG
0001C8I  D100 8138          206           LM     R0,RSAVE2
0001CCI  1800               207           LPSWR  R0
```

```
                                                209  *
                                                210  *
                                                211  *       THE  DISC DRIVER ROUTINE FOR BOTH READ AND WRITE OPERATION
                                                212  *
         0000 01CFI                             213  DRIVER   EQU   *
0001CEI  7350 808E                              214           LHL   CONTER,CONADR      CONTROLLER ADDRESS
0001D2I  DE50 80AA                              215           OC    CONTER,CLEAR
0001D6I  4240 807A                              216           BO    HALT               ABORT IF FALSE SYNC
0001DAI  9D58                                   217  IDLE     SSR   CONTER,COSTAT
0001DCI  2221                                   218           BNPS  IDLE               WAIT IDLE SET
0001DEI  7360 8080                              219  FILE     LHL   DISFIL,FILADR      DISC FILE ADDRESS
0001E2I  9D69                                   220           SSR   DISFIL,DISTAT
0001F4I  4210 8060                              221           RM    HALT               ABORT DISC FILE NOT READY
0001E8I  C390 00C0                              222           THI   DISTAT,X`C0'       WRITE PROTECT, WRITE CHECK ?
0001ECI  4230 8064                              223           BNZ   HALT
0001F0I  C390 0010                              224           THI   DISTAT,X`10'       WAIT FOR INTERLOCK BIT=0
0001F4I  4230 FFE6                              225           BNZ   FILE
0001F8I  C890 00C1                              226           LHI   DISTAT,X`C1'       RESTORE FILE
0001FCI  9E69                                   227           OCR   DISFIL,DISTAT
0001FEI  9D58                                   228  SEERY1   SSR   CONTER,COSTAT
000200I  2221                                   229           BNPS  SEERY1
000202I  9D69                                   230  SEEKZ    SSR   DISFIL,DISTAT
000204I  4270 804C                              231           BTC   7,HALT
000208I  2083                                   232           BCS   SEEKZ
00020AI  73A0 8064                              233           LHL   WORKA,HEAD
00020EI  11A5                                   234           SLLS  WORKA,5
000210I  C4A0 0020                              235           NHI   WORKA,X'20'        HEAD NUMBER 0 OR 1
000214I  73B0 805E                              236           LHL   WORKB,SECTOR
000218I  C4B0 001F                              237           NHI   WORKB,X`1F'
00021CI  C5B0 0018                              238           CLHI  WORKB,X`18'        SECTOR> 23 ?
000220I  4380 8030                              239           BNL   HALT               ABORT
000224I  06AB                                   240           OR    WORKA,WORKB        HEAD+SECTOR
000226I  73C0 804A                              241           LHL   WORKC,CYLNUM       CYLNUM NUMBER
00022AI  C4C0 01FF                              242           NHI   WORKC,X`1FF'
00022EI  07BB                                   243           XR    WORKB,WORKB
000230I  C3C0 0100                              244           THI   WORKC,X`100'
000234I  2332                                   245           BZS   X10
000236I  24B1                                   246           LIS   WORKB,1
000238I  9A6B                                   247  X10      WDR   DISFIL,WORKB       WRITE CYLNUM TO DISC FILE
00023AI  9A6C                                   248           WDR   DISFIL,WORKC
00023CI  DE60 803E                              249           OC    DISFIL,SEEK        ISSUE SEEK COMMAND
000240I  9D58                                   250           SSR   CONTER,COSTAT
000242I  2221                                   251           BFBS  2,1
000244I  9D69                                   252  SEEKLP   SSR   DISFIL,DISTAT
000246I  4270 800A                              253           BTC   7,HALT             EX,SEEK INC, NOT READY ?
00024AI  2083                                   254           BCS   SEEKLP
00024CI  9A6B                                   255           WDR   DISFIL,WORKB       LOAD DISC FILE WITH CYLINDER #
00024EI  9A6C                                   256           WDR   DISFIL,WORKC
000250I  9A5A                                   257           WDR   CONTER,WORKA       LOAD CONTROLLER WIH SECTOR AND HEAD
000252I  030D                                   258           BR    RTN
                                                259  *
                                                260  *
000254I  F810 0000 80F0                         261  HALT     LI    R1,X`80F0'
00025AI  9501                                   262           EPSR  R0,R1
```

ESELCH PROGRAMMING EXAMPLE PAGE 7 14:42:24 03/22/76

```
00025CI  2200                    263             BS      *
00025EI  00F0                    264  SELADR     DCX     F0          SELCH ADDRESS
000260I  00B6                    265  CONADR     DCX     B6          CONTROLLER ADDRESS
000262I  00C6                    266  FILADR     DCX     C6          DISC FILE ADDRESS
000264I  0000                    267  SELCHX     DCX     0           INTERRUPTED SELCH ADDRESS
000266I  0000                    268  SELSTX     DCX     0           INTERRUPTED SELCH STATUS
000268I  0000                    269  CONADRX    DCX     0           INTERRUPTED CONTROLLER ADDRESS
00026AI  0000                    270  CONSTX     DCX     0           INTERRUPTED CONTROLLER STATUS
00026CI  0000                    271  FILADX     DCX     0           INTERRUPTED DISC FILE ADDRESS
00026EI  0000                    272  FILSTX     DCX     0           INTERRUPTED DISC FILE STATUS
000270I  0000                    273  FLAG       DCX     0           BACKGROUND FLAG
000272I  0000                    274  HEAD       DCX     0           HEAD NUMBER
000274I  0000                    275  CYLNUM     DCX     0           CYLINDER NUMBER
000276I  0000                    276  SECTOR     DCX     0           SECTOR NUMBER
000278I  4C                      277  STOP1      DB      X'4C'       EXTENDED ADDRESS,STOP,SELCH STATUS
000279I  48                      278  STOP2      DB      X'48'       STOP,EXTEND ADDRESS
00027AI  74                      279  RDCMD      DB      X'74'       READ MODE,GO
00027BI  54                      280  WRTCMD     DB      X'54'       WRITE MODE,GO
00027CI  41                      281  DREAD      DB      X'41'       READ
00027DI  42                      282  SWRITE     DB      X'42'       WRITE
00027EI  C2                      283  SEEK       DB      X'C2'       FILE SEEK COMMAND
00027FI  40                      284  INC        DB      X'40'
000280I  08                      285  CLEAR      DB      X`08'
000284I                          286  RSAVE      DSF     16
000204I                          287  RSAVE1     DSF     16
000304I                          288  RSAVE2     DSF     16
000344I                          289  INBUF      DSH     128         INPUT BUFFER
         0000 0444I              290  ENDINB     EQU     *
000444I                          291  OUTBUF     DSH     128         OUTPUT BUFFER
         0000 0544I              292  ENDOUT     EQU     *
000544I                          293             END
```

ESELCH PROGRAMMING EXAMPLE PAGE 8 14:42:28 03/22/76

NO ERRORS 0 SQUEZ PASSES

CAL 04-00

ABSTOP	0000 0000												
ADC	0000 0004												
BACKGR	0000 00D4I	122											
CLEAR	0000 0280I	215											
COINT	0000 019EI	93											
CONADR	0000 0260I	52	94	214									
CONADRX	0000 0268I	131	191										
CONSTX	0000 026AI	134	192										
CONTER	0000 0005	52	53	116	132	214	215	217	228	250	257		
CONSTAT	0000 0008	71	72	134	135	217	228	250					
CYINUM	0000 0274I	241											
DISFIL	0000 0006	139	219	220	227	230	247	248	249	252	255	256	
DISINT	0000 01B6I	97											
DISTAT	0000 0009	141	142	220	222	224	226	227	230	252			
DONE	0000 0168I												
DREAD	0000 027CI	116											
DRIVER	0000 01CEI	41	103										
DWRITE	0000 027DI	53											
ENDINB	0000 0444I	105	151										
ENDOUT	0000 0544I	43											
FILADR	0000 0262I	98	219										
FILADX	0000 026CI	137	202										
FILE	0000 01DEI	225											
FILSTX	0000 026EI	141	203										
FINCHK	0000 0124I	138											
FLAG	0000 0270I	88	120	183	194	205							
HALT	0000 0254I	38	62	69	73	102	128	130	133	136	140	143	145	153
		163	216	221	223	231	239	253					
HEAD	0000 0272I	233											
IDLE	0000 01DAI	218											
IMPTOP	0000 0544I												
INBUF	0000 0344I	104	155										
INC	0000 027FI	167											
LADC	0000 0002												
MATCH	0000 0152I	164											
OUTBUF	0000 0444I	42	156										
PURETOP	0000 0000P												
R0	0000 0000	161	162	179	184	185	190	195	196	201	206	207	262
R1	0000 0001	34	35	160	162	261	262						
R14	0000 000E												
R15	0000 000F												
R2	0000 0002	155	160	180	191	202							
R3	0000 0003	35	156	161	181	192	203						
RDCMD	0000 027AI	117											
RSAVE	0000 0284I	179	184										
RSAVE1	0000 02C4I	190	195										
RSAVE2	0000 0304I	201	206										
RTN	0000 000D	41	103	258									
SECTOR	0000 0276I	236											
SEEK	0000 027EI	249											

SEEKLP	0000 0244I	254
SEEKZ	0000 0202I	232
SEERY1	0000 01FEI	229
SELADR	0000 025EI	36 90
SELCH	0000 0004	36 37 39 45 47 49 51 54 57 59 60 63 65
		70 71 101 107 109 111 113 117 127 144 146 148
SELCHX	0000 0264I	126 180
SELINT	0000 0186I	89
SELSTX	0000 0266I	129 181
SENSE	0000 0012I	40
SESTAT	0000 0007	39 57 60 61
START	0000 0000I	174
START2	0000 0070I	
STOP1	0000 0278I	37 59 101 144
STOP2	0000 0279I	70
WAITLP	0000 0040I	58
WORKA	0000 000A	42 44 44 45 46 46 47 63 64 65 67 87 87
		88 89 92 93 96 97 100 104 106 106 107 108 108
		109 118 119 120 121 124 125 126 127 129 131 132 137
		139 146 147 148 150 159 159 160 161 164 166 167 168
		169 171 172 173 182 183 193 194 204 205 233 234 235
		240 257
WORKB	0000 000B	43 48 48 49 50 50 51 68 90 91 92 94 95
		96 98 99 100 105 110 110 111 112 112 113 119 125
		151 152 158 165 168 169 173 236 237 238 240 243 243
		246 247 255
WORKC	0000 000C	64 66 66 67 68 147 149 149 150 152 157 170 170
		171 241 242 244 248 256
WRTCMD	0000 027BI	54
X10	0000 0238I	245

INDEX

PERKIN-ELMER

PUBLICATION COMMENT FORM

We try to make our publications easy to understand and free of errors. Our users are an integral source of information for improving future revisions. Please use this postage paid form to send us comments, corrections, suggestions, *etc.*

1. Publication number_____

2. Title of publication_____

3. Describe, providing page numbers, any technical errors you found. Attach additional sheet if neccessary.

_____

_____

_____

4. Was the publication easy to understand? If no, why not?

_____

5. Were illustrations adequate? _____

_____

_____

6. What additions or deletions would you suggest? _____

_____

_____

7. Other comments: _____

_____

_____

From _____ Date _____

Position/Title _____

Company _____

Address _____

_____

_____

STAPLE

STAPLE

FOLD

FOLD

NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

BUSINESS REPLY MAIL

FIRST CLASS PERMIT NO. 22 OCEANPORT, N.J.

POSTAGE WILL BE PAID BY ADDRESSEE

PERKIN-ELMER

2 Crescent Place
Oceanport, N.J. 07757

ATTN:
TECHNICAL SYSTEMS PUBLICATIONS DEPT., HANCE AVE.

FOLD

FOLD

STAPLE

STAPLE

9306